# $Id: PKGBUILD 165326 2012-08-15 15:47:04Z bisson $
# Contributor: damir <damir@archlinux.org>
# Maintainer: Gaetan Bisson <bisson@archlinux.org>

pkgname=scim-hangul
pkgver=0.4.0
pkgrel=1
pkgdesc='Korean input method module for SCIM'
url='http://www.scim-im.org/'
license=('GPL')
arch=('i686' 'x86_64')
depends=('libhangul' 'scim')
options=('!libtool')
source=("http://downloads.sourceforge.net/sourceforge/scim/${pkgname}-${pkgver}.tar.gz"
        '2012.patch')
sha1sums=('2ba06a6d851a1d86c5a9b659c573bfd87e6a7555'
          'a046c40b78521f85418b8ad2138b51815d47abec')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	patch -p1 -i ../2012.patch
	./configure --prefix=/usr --enable-skim-support
	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
}
